Parameter Estimation for Photovoltaic Diodes

Introduction



The behavior of a photovoltaic diode is often modeled with an equivalent circuit (illustrated above), and is described by the equation below.

$$If = Ipv - I0\left(e^{\frac{If\,Rs + Vf}{n\,Vt}} - 1\right) - \frac{If\,Rs + Vf}{Rp}$$

This application

- will rearrange this equation to give i in terms of the LambertW equation
- find the best-fit parameters against experimental data

References:
http://www.hindawi.com/journals/jam/2013/362619/
Gray, J.L., The Physics of the Solar Cell, in Handbook of Photovoltaic Science and Engineering, A. Luque, Hegedus, S., Editor. 2011, John Wiley and Sons

Rearrange Diode Equation

> *restart* :

> $I\_r := solve\left(If = Ipv - I0\cdot\left(e^{\frac{(Vf + If\cdot Rs)}{n\cdot Vt}} - 1\right) - \frac{Vf + If\cdot Rs}{Rp}, \mathrm{If}\right)$

$$I\_r := -\frac{-\left(-\mathrm{LambertW}\left(-\frac{I0\,Rp\,Rs\,e^{\frac{Rp\,(I0\,Rs + Ipv\,Rs + Vf)}{n\,Vt\,(Rp + Rs)}}}{-Rp\,Vt\,n - Rs\,Vt\,n}\right) + \frac{Rp\,(I0\,Rs + Ipv\,Rs + Vf)}{n\,Vt\,(Rp + Rs)}\right)n\,Vt + Vf}{Rs} \quad \textbf{(2.1}$$

```
> I_pred := unapply(I_r, Vf, Ipv, I0, n, Rs, Rp) :
```

Import Experimental I-V Data for Photo Voltaic Diode

```
> data := ExcelTools:-Import("diode experimental data.xlsx", "Sheet1");
```

data := [1..26 x 1..2 Array
Data Type: anything
Storage: rectangular
Order: Fortran_order] (3.1

```
> V_data := convert(data[.., 1], Vector) :
> I_data := convert(data[.., 2], Vector) :
> p1 := plot(V_data, I_data, style = point, legend = "Experimental Data") :
> T := 273.15 + 33 :
```

$k := 1.380650 \cdot 10^{-23}$:

$q := 1.602176 \cdot 10^{-19}$:

$Vt := \frac{k \cdot T}{q}$:

Find Best-Fit Parameters

```
> res := Statistics:-NonlinearFit(I_pred, V_data, I_data, parameterranges = [0.1..1, 0..0.0001, 1..2, 0.01..0.1, 1..100],
    output = solutionmodule, iterationlimit = 50, optimalitytolerance = 0.01) :
> pars := res:-Results(parametervector);
```

pars := [0.766393737504875
0.00000936308114823049
1.92385643932305
0.0160026044081741
51.3874600644081] (4.1

```
> res:-Results(residualsumofsquares)
```

0.001785240007 (4.2

Plot Model Curve Against Experimental Data

```
> p2 := plot('I_pred'(Vf, pars[1], pars[2], pars[3], pars[4], pars[5]), Vf = min(V_data)..max(V_data), color
    = black, legend = ["Model Curve"], axesfont = [Arial], legendstyle = [font = [Arial]], size = [800, 500], gridlines)
    :
> plots:-display(p1, p2, title = "Parameter Estimation for Photovoltatic Diode", titlefont = [Arial, 18])
```


Parameter Estimation for Photovoltatic Diode
0.7
0.6
0.5
0.4
0.3
0.2
0.1
−0.1
−0.2
−0.2
−0.1
0
0.1
0.2
0.3
0.4
0.5
Experimental Data
Model Curve